3 (e) Detectors with McStas:

Realistic gas detectors and housing effect

E. Farhi <farhi@ill.eu> - Thursday Sept 19th 2013.

References:

Refer to the Workshops and Conferences McStas page at <http://mcstas.org/workshop/>

Scope:

In this exercise, we shall assemble a gas cell (He converter, CF4 stopper), see its intrinsic resolution and efficiency. Then add a material in front (Al or Fe plate) to model its housing. This exercise also uses parameter scanning and the concept of Bragg edge (diffraction).

Simple ideal detectors are usually part of any simulation. Efficiency is 100%. I personally use Monitor_nD.

When a neutron enters a gas cell, it creates at some point a (p,t) pairs. This charge cloud drifts, under electrical field, to a wire where the position is detected e.g. by charge division and coincidence criteria. The PSD_Detector (contributed) component implements most of these aspects.

3E.1: simulating a simple gas cell

Select button **Edit/New.** *The Editor opens*
Select menu **Insert/Template** in
the Editor

Change the instrument **name** as
Exercise3e_detectors and parameter ***lambda=2***

Position the **cursor** after the
TRACE keyword and ***Progress_bar***

Insert a **Source/Source_simple**
and call this instance '*Source*'
Make it a disk of ***radius=0.1*** [mm],
focusing to a 0.1 x 0.1 [mm^2] at 2 m with
neutron wavelength ***lambda0=lambda***
dlambda=0.1



Add an ideal XY monitor at 1.9 m from the Source (5x5 cm², to record all incoming neutrons)

Insert a **Contrib/PSD_Detector**
at 2 [m] from the *PREVIOUS*
component. Make it a BIDIM26 Detector, but with 2.6x2.6x1 cm2
FN_Conv="He3inHe.table",
FN_Stop="He3inCF4.table"

```
COMPONENT Source = Source_simple(
    radius = 0.1e-3, focus_xw = .1e-3, focus_yh = .1e-3,
    lambda0 = lambda)
  AT (0, 0, 0) RELATIVE PREVIOUS

COMPONENT MILAND19 = PSD_Detector(xwidth=0.192, yheight=0.192, nx=64, ny=64,
                    zdepth=0.03, threshold=100, borderx=-1, bordery=-1,
                    PressureConv=5, PressureStop=1,
                    FN_Conv="Gas_tables/He3inHe.table", FN_Stop="Gas_tables/He
                    xChDivRelSigma=0, yChDivRelSigma=0,
                    filename="BIDIM19.psd")
  AT (0,0,2) RELATIVE PREVIOUS
```

Save instrument as 'Exercise3e_detectors.*instr*',
and click the **Run** button.

Select **Trace (3D)** instead of *Simulate*.

Click on the **Start** button.
Look at neutrons inside the detector
zoom with Z key, pass neutrons with SPACE key
Change to Format=PGPLOT and zoom.



Close the Trace view and click again on the **Run** button.
Now select **Simulate** mode with ***Neutron count=1e6***
Start simulation and **Plot** results.
What is the detector resolution and efficiency ?



3E.2: adding an Al housing

Add a 1 [mm] **Al** layer with *reflections="Al.laz"*
in front of the detector, using the *PowderN*
component. This is to model the detector
entry window.

Add a monitor sensitive to scattering in the plate
define a flag to raised when SCATTERED in the plate
then apply WHEN (flag) on the detector.

Launch a single simulation with *lambda=1* and
Plot results. Show Log scale with 'L' key.
Estimate the background from the Al window.



Change the housing plate to iron (Fe.laz).

Estimate the background from the Fe window.
What material should be used to hold the gas ?

Launch a set of simulations scanning with the *Al.laz* plate
lambda=.2,5 [Angs] in 20 steps.

Plot results [Ideal total | scattered on housing | BIDIM].
Comments about the gas detector efficiency ?
Comments about the scattered signal from the housing ?



Solution to 3E.2:

```
/*******************************************************************************
*         McStas instrument definition URL=http://www.mcstas.org
*
* Instrument: Exercise3e_detectors
*
* %Identification
* Written by: Luke Skywalker
* Origin: The Moon
* Release: McStas CVS-080208
* Version: 0.2
* %INSTRUMENT_SITE: Death_Star
*
* A simple detector that shows many features
*
* %Description
* A gas cell, with a housing plate in front. Beam is a single pencil-like ray.
*
* Example: mcrun Exercise3e_detectors.instr lambda=2
*
* %Parameters
* lambda: [Angs]   Incoming wavelength
* window: [string] material used as housing
*
* %End
*******************************************************************************/

DEFINE INSTRUMENT Exercise3e_detectors(lambda=2, string window="Al.laz")

DECLARE %{
  int flag_window=0;
%}

TRACE

COMPONENT Origin = Progress_bar()
  AT (0,0,0) ABSOLUTE
  EXTEND %{
    flag_window=0;
  %}
COMPONENT Source = Source_simple(
    radius = 1e-4, dist = 2, focus_xw = 0.1e-3, focus_yh = 0.1e-3,
    lambda0 = lambda, dlambda = 0.1)
  AT (0, 0, 0) RELATIVE Origin

COMPONENT Window = PowderN(
    xwidth=0.1, yheight=0.1, zdepth=0.001,
    reflections=window, p_transmit=0.9, p_inc=0.05
  )
  AT (0, 0, 2-0.02) RELATIVE Source
  EXTEND %{
    if (SCATTERED) flag_window=1;
  %}

COMPONENT PSD_total=Monitor_nD(bins=128, xwidth=0.05, yheight=0.05)
  AT(0,0,2.01) RELATIVE Source
COMPONENT PSD_window=COPY(PSD_total)
  WHEN (flag_window)
  AT(0,0,2.02) RELATIVE Source

COMPONENT BIDIM26=PSD_Detector(xwidth=0.026, yheight=0.026, nx=128, ny=128,
  zdepth=0.03, PressureConv=5, PressureStop=1,
  FN_Conv="Gas_tables/He3inHe.table", FN_Stop="Gas_tables/He3inCF4.table",
  filename="BIDIM26.psd")
  AT(0,0,2.03) RELATIVE Source

END
```